Otro pescado

1. podemos utilizar el proyecto anterior y continuar desde hay para ello debemos crearlo nuevo para no dañarlo los pasos son estos por si no se acuerdan del proyecto

2. Cree un proyecto vacío y repita todo el proceso del proyecto anterior para comenzar este desde hay, no se usa el anterior porque puede que quede código mal colocado o alguna imagen mal usada y se daña el proyecto además de que se puede intentar hacerlo sin seguir instrucciones lo que hace que podamos crearlo sin estar mirando las instrucciones solo con poner los scripts en la carpeta de este proyecto lo demás lo hacemos de memoria y verificamos que ya podemos crear aspectos básicos nosotros mismos

3. lo que haremos es crear un nuevo script llamado Fish.cs y colocamos lo siguiente.
function Fish::getSpeed(%this)

{

   return getRandom(%this.minSpeed, %this.maxSpeed);

}

function Fish::reposition(%this)

{

   %this.setPositionY(getRandom(-45, 30));

}

function Fish::onLevelLoaded(%this, %scenegraph)

{

   %this.setLinearVelocityX(%this.getSpeed());

}

function Fish::onWorldLimit(%this, %mode, %limit)

{

   switch$(%limit)

   {

      case "left":

         %this.setFlipX(false);

         %this.setLinearVelocityX(%this.getSpeed());

         %this.reposition();

      case "right":

         %this.setFlipX(true);

         %this.setLinearVelocityX(-%this.getSpeed());

         %this.reposition();

   }

}

4. agregue la llamada desde el game.cs así: exec("./fish.cs"); guarde y cierre ambos archivos

1. <Create> + (Animated Sprites) arrastre “Angelfish2SheetImageMapAnimation”
2. <Edit> + (Scene Object) + (Layer) = 16
5. <Edit> + (Scripting) + (Class) = Fish
6. <Edit> + (World limits) + (Limit mode) = “NULL” y Check el “CallBack”

7. Cambie los limites del mundo del nuevo pez haga que rodee toda la vista del escenario y aumente hacia la izquierda y derecha el limite mas que de arriba y abajo así:

8. ubique el pez en la esquina derecha superior

9. con el pez nuevo seleccionado realice lo siguiente:
10. <Edit> + (Collision) +1 (Receive Collision) = check, +2 (Detection Mode) = FULL, +3 (Collision Response) = CLAMP y a los demás dejarlos sin check
11. <Edit> + (Dinamic Fields) + (maxSpeed) = 30 y luego agregue [image: image2.png]


 otro cuadro y escriba (minSpeed) = 5 luego agregue [image: image3.png]


 otro cuadro para terminar
12. ahora vamos a adicionar mas pescados lo que hacemos es muy facil solo es copiar nuestro pez enemigo y copiarlo varias veces mas, como al pegar queda encima del original debe moverse a otra parte de la escena para copiarlo debe hacerse desde el menú principal con la opción <Edit> + (Copy) y (Paste)
13. ahora seleccione uno de los peces que agrego y si no queremos enemigos iguales los podemos animar con la imagen de otro pez así:
<Edit> + (Animated Sprite) + “Animation” = seleccione la imagen del pez que quiera ver como reemplazo del enemigo que seleccionó, puede hacerlo para cada uno de los enemigos
14. No se olvide que cada animación de peces tiene una variación de la ubicación de la boca y por ello hay que modificar eso con la herramienta [image: image4.png]@D0E

Edt ihis obiect




15. como se había hecho en el proyecto anterior se debe colocar un polígono que delimite la colisión de la burbuja con la boca del pez

16. ahora debemos crear un marcador de vida de nuestro pez (jugador)

17. seleccionamos nuestro pez (jugador) y realizamos lo siguiente:

18. <Edit> + (Dinamic Fields) 
+ (lifeDrain) = -2 luego agregue [image: image5.png]


  (este item es el nivel de dificultad -1 como facil y -4 como el mas difícil no altere mas este item o perderá el sentido de la disminución o simplemente no pasará nada es decir nunca muere el pez o morirá demasiado rapido)
+ (life) = 100 luego agregue [image: image6.png]


 

+ (maxHeight) = 20 luego agregue [image: image7.png]



+ (maxWidth) = 18 luego agregue [image: image8.png]


 otro cuadro para terminar
19. coloque este código en el archivo placer.cs compare o borre todo el codigo y reemplace por este
function PlayerFish::onLevelLoaded(%this, %scenegraph)

{

   $FishPlayer = %this;

   moveMap.bindCmd(keyboard, "Up", "fishPlayerUp();", "fishPlayerUpStop();");

   moveMap.bindCmd(keyboard, "Down", "fishPlayerDown();", "fishPlayerDownStop();");

   moveMap.bindCmd(keyboard, "Left", "fishPlayerLeft();", "fishPlayerLeftStop();");

   moveMap.bindCmd(keyboard, "Right", "fishPlayerRight();", "fishPlayerRightStop();");

   moveMap.bindCmd(keyboard, "space", "fishPlayerBoost();", "fishPlayerBoostStop();");

   %this.lowerLife();

}

function PlayerFish::updateMovement(%this)

{

   if(%this.dead)

      return;

   if(%this.moveLeft)

   {

      $FishPlayer.setFlipX(true);

      $FishPlayer.setLinearVelocityX( -$FishPlayer.hSpeed );

   }

   if(%this.moveRight)

   {

      $FishPlayer.setFlipX(false);

      $FishPlayer.setLinearVelocityX( $FishPlayer.hSpeed );

   }

   if(%this.moveUp)

   {

      %this.setLinearVelocityY( -$FishPlayer.vSpeed );

   }

   if(%this.moveDown)

   {

      %this.setLinearVelocityY( $FishPlayer.vSpeed );

   }

   if(!%this.moveLeft && !%this.moveRight)

   {

      %this.setLinearVelocityX( 0 );

   }

   if(!%this.moveUp && !%this.moveDown)

   {

      %this.setLinearVelocityY( 0 );

   }

}

function PlayerFish::modifyLife(%this, %dmg)

{

   %this.life += %dmg;

   if(%this.life > 100)

   {

      %this.life = 100;

   } else if (%this.life < 0)

   {

      %this.life = 0;

   }

   if(%this.life <= 30)

   {

      %this.dead();

   } else

   {

      %this.updateLifeSize();

   }

}

function PlayerFish::updateLifeSize(%this)

{

   %lifeMultiplier = %this.life / 100;

   %newWidth = %this.maxWidth * %lifeMultiplier;

   %newHeight = %this.maxHeight * %lifeMultiplier;

   %this.setSize(%newWidth, %newHeight);

}

function PlayerFish::dead(%this)

{

   %this.setFlipY(true);

   %this.setLinearVelocityY(-10);

   %this.dead = true;

}

function PlayerFish::lowerLife(%this)

{

   %this.modifyLife(%this.lifeDrain);

   if(!%this.dead)

   {

      %this.schedule(500, "lowerLife");

   }

}

function fishPlayerUp()

{

   $FishPlayer.moveUp = true;

   $FishPlayer.updateMovement();

}

function fishPlayerDown()

{

   $FishPlayer.moveDown = true;

   $FishPlayer.updateMovement();

}

function fishPlayerLeft()

{

   $FishPlayer.moveLeft = true;

   $FishPlayer.updateMovement();

}

function fishPlayerRight()

{

   $FishPlayer.moveRight = true;

   $FishPlayer.updateMovement();

}

function fishPlayerUpStop()

{

   $FishPlayer.moveUp = false;

   $FishPlayer.updateMovement();

}

function fishPlayerDownStop()

{

   $FishPlayer.moveDown = false;

   $FishPlayer.updateMovement();

}

function fishPlayerLeftStop()

{

   $FishPlayer.moveLeft = false;

   $FishPlayer.updateMovement();

}

function fishPlayerRightStop()

{

   $FishPlayer.moveRight = false;

   $FishPlayer.updateMovement();

}

function fishPlayerBoost()

{

   if($FishPlayer.dead)

      return;   

   %flipX = $FishPlayer.getFlipX();

   if(!%flipX)

   {

      %hSpeed = $FishPlayer.hSpeed * 3;

   } else

   {

      %hSpeed = -$FishPlayer.hSpeed * 3;

   }

   $FishPlayer.setLinearVelocityX(%hSpeed);

}

function fishPlayerBoostStop()

{

   $FishPlayer.setLinearVelocityX(0);

}

20. y cambie este código en el archivo de FishFood.cs 

function FishFood::onCollision(%srcObj, %dstObj, %srcRef, %dstRef, %time, %normal, %contactCount, %contacts)

{

   if(%dstObj.class $= "PlayerFish")

   {

      %srcObj.spawn();

      %dstObj.modifyLife(%srcObj.lifeValue);

   } else if (%dstObj.class $= "Fish")

   {

      %srcObj.spawn();

   }

}

21.  seleccione la burbuja y adicione esta opción <Edit> + (Dinamic Fields) + (lifeValue) = 10
